The Stochastic Power Law Logistic Model: Derivations of Asymptotic Approximations of the first 3 Cumulants of the QSD, with s=8.

Ingemar Nåsell

The procedure **cumf** is used to determine the cumulant of order kk as a function of the (raw) moments of orders up to kk, while the procedure **muf** determines the (raw) moment of order kk as a function of the cumulants of orders up to kk. The procedure **ODEcum3** derives ODEs for the first 3 cumulants.
More details are given in the Maple work-sheet DeriveODECum3.

```
> restart;
```

```
> cumf:=proc(kk)
   local M,K,Ks,cum;
   description "Determines the cumulant of order kk as function of
  the raw moments of orders up to kk";
   M:=1+add(cat(mu,k)*theta^k/k!,k=1..kk);
   K:=log(M);
   Ks:=convert(series(K,theta,kk+1),polynom);
   cum:=sort(simplify(coeff(Ks,theta,kk)*kk!),[seq(cat(mu,j),j=1..
  kk)],plex);
  end proc:
```

```
> muf:=proc(kk)
   local K,M,Ms,mu;
   description "Determines the (raw) moment of order kk as function
  of the cumulants of orders up to kk";
   K:=add(cat(kappa,k)*theta^k/k!,k=1..kk);
   M:=exp(K);
   Ms:=convert(series(M,theta,kk+1),polynom);
   mu:=sort(coeff(Ms,theta,kk)*kk!,[seq(cat(kappa,j),j=1..kk)],
  plex);
  end proc:
```

```
> ODEcum3:=proc(s)
    local Dmu1,Dmu2,Dmu3,sub1,sub2,Dcum1,Dcum2,Dcum3,a,c,d,b;
    description "Derive ODEs for the first 3 cumulants";
    Dmu1:=a*mu1-b*cat(mu,s+1);
    Dmu2:=2*a*mu2+c*mu1-d*cat(mu,s+1)-2*b*cat(mu,s+2);
    Dmu3:=a*(mu1+3*mu3)+3*c*mu2-3*d*cat(mu,s+2)-b*(cat(mu,s+1)+3*
  cat(mu,s+3));
    sub1:=D(mu1)=Dmu1,D(mu2)=Dmu2,D(mu3)=Dmu3;
    sub2:=seq(cat(mu,i)=muf(i),i=1..3+s);
    Dcum1:=map(sort,map(simplify,collect(subs(sub2,subs(sub1,D(cumf
  (1)))),[a,c,d,b])));
```

```
  Dcum2:=map(sort,map(simplify,collect(subs(sub2,subs(sub1,D(cumf
(2)))),[a,c,d,b])));
  Dcum3:=map(sort,map(simplify,collect(subs(sub2,subs(sub1,D(cumf
(3)))),[a,c,d,b])));
  a:=mu*(R0-1);
  c:=mu*(R0+1);
  d:=mu*(R0-alpha)/N^s;
  b:=mu*(R0+alpha)/N^s;
  [eval(Dcum1),eval(Dcum2),eval(Dcum3)];
end proc:
```

Put s=8 and denote the derivatives of the first 3 cumulants by A, B, C.

```
> s:=8;
```

$$s := 8 \tag{1}$$

```
> A:=ODEcum3(s)[1];
  B:=ODEcum3(s)[2];
  C:=ODEcum3(s)[3];
```

$$A := \mu\,(R0-1)\,\kappa 1 - \frac{1}{N^8}\Big(\big(\kappa 1^9 + 36\,\kappa 1^7\,\kappa 2 + 84\,\kappa 1^6\,\kappa 3 + \big(378\,\kappa 2^2 + 126\,\kappa 4\big)\,\kappa 1^5 + (1260\,\kappa 2\,\kappa 3 + 126\,\kappa 5)\,\kappa 1^4 + 1260\,\kappa 2^3\,\kappa 3 + \big(1260\,\kappa 2^3 + 1260\,\kappa 2\,\kappa 4 + 840\,\kappa 3^2 + 84\,\kappa 6\big)\,\kappa 1^3 + 378\,\kappa 2^2\,\kappa 5 + 280\,\kappa 3^3 + \big(3780\,\kappa 2^2\,\kappa 3 + 756\,\kappa 2\,\kappa 5 + 1260\,\kappa 3\,\kappa 4 + 36\,\kappa 7\big)\,\kappa 1^2 + 84\,\kappa 3\,\kappa 6 + 126\,\kappa 4\,\kappa 5 + \big(945\,\kappa 2^4 + 1890\,\kappa 2^2\,\kappa 4 + 504\,\kappa 3\,\kappa 5 + 315\,\kappa 4^2 + \big(2520\,\kappa 3^2 + 252\,\kappa 6\big)\,\kappa 2 + 9\,\kappa 8\big)\,\kappa 1 + (1260\,\kappa 3\,\kappa 4 + 36\,\kappa 7)\,\kappa 2 + \kappa 9\big)\,\mu\,(R0+\alpha)\Big)$$

$$B := 2\,\mu\,(R0-1)\,\kappa 2 + \mu\,(R0+1)\,\kappa 1 - \frac{1}{N^8}\Big(\big(\kappa 1^9 + 36\,\kappa 1^7\,\kappa 2 + 84\,\kappa 1^6\,\kappa 3 + \big(378\,\kappa 2^2 + 126\,\kappa 4\big)\,\kappa 1^5 + (1260\,\kappa 2\,\kappa 3 + 126\,\kappa 5)\,\kappa 1^4 + 1260\,\kappa 2^3\,\kappa 3 + \big(1260\,\kappa 2^3 + 1260\,\kappa 2\,\kappa 4 + 840\,\kappa 3^2 + 84\,\kappa 6\big)\,\kappa 1^3 + 378\,\kappa 2^2\,\kappa 5 + 280\,\kappa 3^3 + \big(3780\,\kappa 2^2\,\kappa 3 + 756\,\kappa 2\,\kappa 5 + 1260\,\kappa 3\,\kappa 4 + 36\,\kappa 7\big)\,\kappa 1^2 + 84\,\kappa 3\,\kappa 6 + 126\,\kappa 4\,\kappa 5 + \big(945\,\kappa 2^4 + 1890\,\kappa 2^2\,\kappa 4 + 504\,\kappa 3\,\kappa 5 + 315\,\kappa 4^2 + \big(2520\,\kappa 3^2 + 252\,\kappa 6\big)\,\kappa 2 + 9\,\kappa 8\big)\,\kappa 1 + (1260\,\kappa 3\,\kappa 4 + 36\,\kappa 7)\,\kappa 2 + \kappa 9\big)\,\mu\,(R0-\alpha)\Big) - \frac{1}{N^8}\Bigg(18\,\Bigg(\kappa 1^8\,\kappa 2 + 4\,\kappa 1^7\,\kappa 3 + \left(28\,\kappa 2^2 + \frac{28\,\kappa 4}{3}\right)\kappa 1^6 + (140\,\kappa 2\,\kappa 3 + 14\,\kappa 5)\,\kappa 1^5 + 105\,\kappa 2^5 + \big(210\,\kappa 2^3 + 210\,\kappa 2\,\kappa 4 + 140\,\kappa 3^2 + 14\,\kappa 6\big)\,\kappa 1^4 + 350\,\kappa 2^3\,\kappa 4 + \left(980\,\kappa 2^2\,\kappa 3 + 196\,\kappa 2\,\kappa 5 + \frac{980}{3}\,\kappa 3\,\kappa 4 + \frac{28}{3}\,\kappa 7\right)\kappa 1^3 + \frac{700\,\kappa 3^2\,\kappa 4}{3} + \big(420\,\kappa 2^4 + 840\,\kappa 2^2\,\kappa 4 + 224\,\kappa 3\,\kappa 5 + 140\,\kappa 4^2 + \big(1120\,\kappa 3^2$$

$$+ 112\,\kappa 6)\ \kappa 2 + 4\,\kappa 8)\ \kappa 1^{2} + (700\,\kappa 3^{2} + 70\,\kappa 6)\ \kappa 2^{2} + \frac{40\,\kappa 3\,\kappa 7}{3} + \frac{70\,\kappa 4\,\kappa 6}{3}$$

$$+ 14\,\kappa 5^{2} + (1260\,\kappa 2^{3}\,\kappa 3 + 378\,\kappa 2^{2}\,\kappa 5 + 280\,\kappa 3^{3} + 84\,\kappa 3\,\kappa 6 + 126\,\kappa 4\,\kappa 5$$

$$+ (1260\,\kappa 3\,\kappa 4 + 36\,\kappa 7)\ \kappa 2 + \kappa 9)\ \kappa 1 + \frac{\kappa 10}{9} + (280\,\kappa 3\,\kappa 5 + 175\,\kappa 4^{2}$$

$$+ 5\,\kappa 8)\ \kappa 2)\ \mu\,(R0 + \alpha))$$

$$C := (\kappa 1 + 3\,\kappa 3)\ \mu\,(R0 - 1) + 3\,\mu\,(R0 + 1)\ \kappa 2 - \frac{1}{N^{8}}(27\,(\kappa 1^{8}\,\kappa 2 + 4\,\kappa 1^{7}\,\kappa 3 \qquad \mathbf{(2)}$$

$$+ \left(28\,\kappa 2^{2} + \frac{28\,\kappa 4}{3}\right) \kappa 1^{6} + (140\,\kappa 2\,\kappa 3 + 14\,\kappa 5)\ \kappa 1^{5} + 105\,\kappa 2^{5} + (210\,\kappa 2^{3}$$

$$+ 210\,\kappa 2\,\kappa 4 + 140\,\kappa 3^{2} + 14\,\kappa 6)\ \kappa 1^{4} + 350\,\kappa 2^{3}\,\kappa 4 + (980\,\kappa 2^{2}\,\kappa 3 + 196\,\kappa 2\,\kappa 5$$

$$+ \frac{980}{3}\,\kappa 3\,\kappa 4 + \frac{28}{3}\,\kappa 7)\ \kappa 1^{3} + \frac{700\,\kappa 3^{2}\,\kappa 4}{3} + (420\,\kappa 2^{4} + 840\,\kappa 2^{2}\,\kappa 4$$

$$+ 224\,\kappa 3\,\kappa 5 + 140\,\kappa 4^{2} + (1120\,\kappa 3^{2} + 112\,\kappa 6)\ \kappa 2 + 4\,\kappa 8)\ \kappa 1^{2} + (700\,\kappa 3^{2}$$

$$+ 70\,\kappa 6)\ \kappa 2^{2} + \frac{40\,\kappa 3\,\kappa 7}{3} + \frac{70\,\kappa 4\,\kappa 6}{3} + 14\,\kappa 5^{2} + (1260\,\kappa 2^{3}\,\kappa 3 + 378\,\kappa 2^{2}\,\kappa 5$$

$$+ 280\,\kappa 3^{3} + 84\,\kappa 3\,\kappa 6 + 126\,\kappa 4\,\kappa 5 + (1260\,\kappa 3\,\kappa 4 + 36\,\kappa 7)\ \kappa 2 + \kappa 9)\ \kappa 1 + \frac{\kappa 10}{9}$$

$$+ (280\,\kappa 3\,\kappa 5 + 175\,\kappa 4^{2} + 5\,\kappa 8)\ \kappa 2)\ \mu\,(R0 - \alpha)) - \frac{1}{N^{8}}(108\,(\frac{\kappa 1^{9}}{108}$$

$$+ \frac{\kappa 1^{8}\,\kappa 3}{4} + \left(2\,\kappa 2^{2} + \frac{1}{3}\,\kappa 2 + \kappa 4\right) \kappa 1^{7} + \left(21\,\kappa 2\,\kappa 3 + \frac{7}{9}\,\kappa 3 + \frac{7}{3}\,\kappa 5\right) \kappa 1^{6}$$

$$+ \left(42\,\kappa 2^{3} + \frac{7}{2}\,\kappa 2^{2} + 49\,\kappa 2\,\kappa 4 + 35\,\kappa 3^{2} + \frac{7}{6}\,\kappa 4 + \frac{7}{2}\,\kappa 6\right) \kappa 1^{5} + \frac{1785\,\kappa 2^{4}\,\kappa 3}{4}$$

$$+ \left(\frac{665\,\kappa 2^{2}\,\kappa 3}{2} + \frac{245\,\kappa 3\,\kappa 4}{2} + \left(\frac{35\,\kappa 3}{3} + 70\,\kappa 5\right) \kappa 2 + \frac{7\,\kappa 5}{6} + \frac{7\,\kappa 7}{2}\right) \kappa 1^{4}$$

$$+ (210\,\kappa 2^{4} + \frac{35\,\kappa 2^{3}}{3} + 455\,\kappa 2^{2}\,\kappa 4 + \frac{70\,\kappa 3^{2}}{9} + \frac{392\,\kappa 3\,\kappa 5}{3} + \frac{245\,\kappa 4^{2}}{3}$$

$$+ \left(630\,\kappa 3^{2} + \frac{35\,\kappa 4}{3} + 63\,\kappa 6\right) \kappa 2 + \frac{7\,\kappa 6}{9} + \frac{7\,\kappa 8}{3})\ \kappa 1^{3} + (\frac{35\,\kappa 3}{3}$$

$$+ 182\,\kappa 5)\ \kappa 2^{3} + \frac{70\,\kappa 3^{3}}{27} + \frac{385\,\kappa 3^{2}\,\kappa 5}{3} + (1155\,\kappa 2^{3}\,\kappa 3 + 280\,\kappa 3^{3} + (35\,\kappa 3$$

$$+357\,\kappa 5)\,\kappa 2^2+126\,\kappa 4\,\kappa 5+(1225\,\kappa 3\,\kappa 4+7\,\kappa 5+35\,\kappa 7)\,\kappa 2+\left(\frac{35\,\kappa 4}{3}+84\,\kappa 6\right)\kappa 3+\frac{\kappa 7}{3}+\kappa 9\right)\kappa 1^2+\left(\frac{1855\,\kappa 3\,\kappa 4}{2}+\frac{7\,\kappa 5}{2}+\frac{53\,\kappa 7}{2}\right)\kappa 2^2$$

$$+\frac{77\,\kappa 5\,\kappa 6}{6}+\left(210\,\kappa 2^5+\frac{35\,\kappa 2^4}{4}+735\,\kappa 2^3\,\kappa 4+525\,\kappa 3^2\,\kappa 4+\left(1505\,\kappa 3^2+\frac{35\,\kappa 4}{2}+\frac{301\,\kappa 6}{2}\right)\kappa 2^2+\frac{35\,\kappa 4^2}{12}+\frac{105\,\kappa 4\,\kappa 6}{2}+\frac{63\,\kappa 5^2}{2}+\frac{\kappa 10}{4}+\left(\frac{70}{3}\,\kappa 3^2\right.\right.$$

$$\left.\left.+616\,\kappa 3\,\kappa 5+385\,\kappa 4^2+\frac{7}{3}\,\kappa 6+11\,\kappa 8\right)\kappa 2+\left(\frac{14\,\kappa 5}{3}+30\,\kappa 7\right)\kappa 3+\frac{\kappa 8}{12}\right)\kappa 1$$

$$+\frac{\kappa 11}{36}+\left(420\,\kappa 3^3+189\,\kappa 4\,\kappa 5+\left(\frac{35\,\kappa 4}{3}+126\,\kappa 6\right)\kappa 3+\frac{\kappa 7}{3}+\frac{3\,\kappa 9}{2}\right)\kappa 2$$

$$+\left(\frac{1925\,\kappa 4^2}{12}+\frac{7\,\kappa 6}{9}+\frac{55\,\kappa 8}{12}\right)\kappa 3+\left(\frac{7\,\kappa 5}{6}+\frac{55\,\kappa 7}{6}\right)\kappa 4+\frac{\kappa 9}{108}\right)\mu\,(R0+\alpha)\Big)$$

Determine asymptotic approximations of the first 3 cumulants of the following forms. Assume also that kappa4 - kappa11 are O(N).

```
> kappa1:=x1*N + x2 + x3/N:
  kappa2:=y1*N + y2:
  kappa3:=z1*N:
  kappa4:=u1*N:
  kappa5:=u2*N:
  kappa6:=u3*N:
  kappa7:=u4*N:
  kappa8:=u5*N:
  kappa9:=u6*N:
  kappa10:=u7*N:
  kappa11:=u8*N:
```

Asymptotic expressions for A, B, C are written A = A1*N + A2 + A3/N, B = B1*N + B2, C = C1*N, where

```
> A1:=coeff(A,N,1); A2:=coeff(A,N,0); A3:=coeff(A,N,-1);
  B1:=coeff(B,N,1); B2:=coeff(B,N,0);
  C1:=coeff(C,N,1);
```

$$A1 := \mu\,(R0-1)\,x1 - x1^9\,\mu\,(R0+\alpha)$$

$$A2 := \mu\,(R0-1)\,x2 - (9\,x2\,x1^8+36\,x1^7\,y1)\,\mu\,(R0+\alpha)$$

$$A3 := \mu\,(R0-1)\,x3 - (x3\,x1^8+8\,x2^2\,x1^7+x1\,(2\,(2\,(2\,x3\,x1+x2^2)\,x1^2+4\,x2^2\,x1^2)\,x1^4+16\,x2^2\,x1^6)+36\,x1^7\,y2+252\,x2\,x1^6\,y1+84\,x1^6\,z1+378\,y1^2\,x1^5)\,\mu\,(R0+\alpha)$$

$$B1 := 2\,\mu\,(R0-1)\,y1+\mu\,(R0+1)\,x1-x1^{9}\,\mu\,(R0-\alpha)-18\,x1^{8}\,y1\,\mu\,(R0+\alpha)$$
$$B2 := 2\,\mu\,(R0-1)\,y2+\mu\,(R0+1)\,x2-(9\,x1^{8}\,x2+36\,x1^{7}\,y1)\,\mu\,(R0-\alpha) - 18\,(x1^{8}\,y2+8\,x2\,x1^{7}\,y1+4\,x1^{7}\,z1+28\,y1^{2}\,x1^{6})\,\mu\,(R0+\alpha)$$
$$C1 := (x1+3\,z1)\,\mu\,(R0-1)+3\,\mu\,(R0+1)\,y1-27\,x1^{8}\,y1\,\mu\,(R0-\alpha) - 108\left(\frac{1}{108}\,x1^{9}+\frac{1}{4}\,x1^{8}\,z1+2\,y1^{2}\,x1^{7}\right)\mu\,(R0+\alpha) \tag{3}$$

6 equations can now be formed by setting each of A1, A2, A3, B1, B2, C1 equal to zero. These equations can be solved for the 6 unknowns x1, x2, x3, y1, y2, z1. The solutions are found sequentially, as follows:
First we solve A1=0 for x1.This equation has s+1=9 solutions. Among them, we exclude s=8 spurious solution.
After this, B1=0 is solved for for y1, A2=0 is solved for x2, C1=0 is solved for z1, B2=0 is solved for y2, and A3=0 is solved for x3.

```
> x1a:=solve(A1,x1);
  x1:=((R0-1)/(R0+alpha))^(1/8);
```

$$x1a := 0,\ \frac{\left((R0-1)\,(R0+\alpha)^{7}\right)^{1/8}}{R0+\alpha},\ \frac{\left(\frac{\sqrt{2}}{2}+\frac{I\sqrt{2}}{2}\right)\left((R0-1)\,(R0+\alpha)^{7}\right)^{1/8}}{R0+\alpha},$$
$$\frac{I\left((R0-1)\,(R0+\alpha)^{7}\right)^{1/8}}{R0+\alpha},\ \frac{\left(-\frac{\sqrt{2}}{2}+\frac{I\sqrt{2}}{2}\right)\left((R0-1)\,(R0+\alpha)^{7}\right)^{1/8}}{R0+\alpha},$$
$$-\frac{\left((R0-1)\,(R0+\alpha)^{7}\right)^{1/8}}{R0+\alpha},\ \frac{\left(-\frac{\sqrt{2}}{2}-\frac{I\sqrt{2}}{2}\right)\left((R0-1)\,(R0+\alpha)^{7}\right)^{1/8}}{R0+\alpha},$$
$$\frac{-I\left((R0-1)\,(R0+\alpha)^{7}\right)^{1/8}}{R0+\alpha},\ \frac{\left(\frac{\sqrt{2}}{2}-\frac{I\sqrt{2}}{2}\right)\left((R0-1)\,(R0+\alpha)^{7}\right)^{1/8}}{R0+\alpha}$$
$$x1 := \left(\frac{R0-1}{R0+\alpha}\right)^{1/8} \tag{4}$$

```
> y1:=factor(solve(B1,y1));
```

$$y1 := \frac{\left(\frac{R0-1}{R0+\alpha}\right)^{1/8} R0\,(1+\alpha)}{8\,(R0+\alpha)\,(R0-1)} \tag{5}$$

```
> x2:=solve(A2,x2);
```

$$x2 := -\frac{9\,R0\,(1+\alpha)}{16\,(R0+\alpha)\,(R0-1)} \tag{6}$$

```
> z1:=factor(solve(C1,z1));
```

(7)

$$z1 := -\frac{RO\left(\frac{RO-1}{RO+\alpha}\right)^{1/8}(1+\alpha)\,(8\,RO^2-\alpha\,RO-RO+8\,\alpha)}{64\,(RO+\alpha)^2\,(RO-1)^2} \tag{7}$$

```
> y2:=solve(B2,y2);
```

$$y2 := \frac{9\,(RO^2+\alpha)\,RO\,(1+\alpha)}{16\,(RO+\alpha)^2\,(RO-1)^2} \tag{8}$$

```
> x3:=factor(solve(A3,x3));
```

$$x3 := -\frac{3\left(\frac{RO-1}{RO+\alpha}\right)^{7/8}RO\,(1+\alpha)\,(208\,RO^2+19\,\alpha\,RO+19\,RO+208\,\alpha)}{512\,(RO-1)^3\,(RO+\alpha)} \tag{9}$$

Summarize the reslts for s=8:

```
> x1, x2, x3, y1, y2, z1;
```

$$\left(\frac{RO-1}{RO+\alpha}\right)^{1/8}, -\frac{9\,RO\,(1+\alpha)}{16\,(RO+\alpha)\,(RO-1)}, -\frac{3\left(\frac{RO-1}{RO+\alpha}\right)^{7/8}RO\,(1+\alpha)\,(208\,RO^2+19\,\alpha\,RO+19\,RO+208\,\alpha)}{512\,(RO-1)^3\,(RO+\alpha)}, \frac{\left(\frac{RO-1}{RO+\alpha}\right)^{1/8}RO\,(1+\alpha)}{8\,(RO+\alpha)\,(RO-1)}, \frac{9\,(RO^2+\alpha)\,RO\,(1+\alpha)}{16\,(RO+\alpha)^2\,(RO-1)^2}, -\frac{RO\left(\frac{RO-1}{RO+\alpha}\right)^{1/8}(1+\alpha)\,(8\,RO^2-\alpha\,RO-RO+8\,\alpha)}{64\,(RO+\alpha)^2\,(RO-1)^2} \tag{10}$$

```
>
```